z-filtered 3Q MQMAS




```
;mp3qzf

1 ze
2 d1
  3u pl11:f1
  (3u ph1):f1
  (p1 ph1):f1

if "l0==1"{
"d0=in0-1u-p1/2-p2/2"
}
if "l0>0"{
  d0
}
  1u

  (p2 ph2):f1
  d4
  2.5u pl13:f1
  (p3 ph3):f1
  go=2 ph31
  10m mc #0 to 2 F1PH(ip1,id0&iu0)
exit

ph1=(12)0 2 4 6 8 10
ph2=0
ph3={0}*6 {1}*6 {2}*6 {3}*6
ph0=0
ph31={0 2 0 2 0 2}^1^2^3
```

Energy levels and coherence orders



	$\lvert +1/2\rangle$	$\lvert -1/2\rangle$
$\langle +1/2\rvert$	0Q	1Q
$\langle -1/2\rvert$	1Q	0Q



	$\lvert +3/2\rangle$	$\lvert +1/2\rangle$	$\lvert -1/2\rangle$	$\lvert -3/2\rangle$
$\langle +3/2\rvert$	0Q	1Q ST	2Q	3Q
$\langle +1/2\rvert$	1Q ST	0Q	1Q CT	2Q
$\langle -1/2\rvert$	2Q	1Q CT	0Q	1Q ST
$\langle -3/2\rvert$	3Q	2Q	1Q ST	0Q



	$\lvert \alpha\alpha\rangle$	$\lvert \alpha\beta\rangle$	$\lvert \beta\alpha\rangle$	$\lvert \beta\beta\rangle$
$\langle \alpha\alpha\rvert$	0Q	1Q	1Q	2Q
$\langle \alpha\beta\rvert$	1Q	0Q	0Q	1Q
$\langle \beta\alpha\rvert$	1Q	0Q	0Q	1Q
$\langle \beta\beta\rvert$	2Q	1Q	1Q	0Q

One-dimensional pulse sequences




```
;zg

1 ze
2 d1
  3u pl1:f1
  (p1 ph1):f1
  go=2 ph31
  10m wr #0
exit

ph1= 0 2 1 3
ph31=0 2 1 3
```

- phase cycle to remove ringing/pulse breakthrough, quadrature images, transmitter/DC offset glitch, which are independent of pulse phase




```
;se

1 ze
2 d1
  3u pl1:f1
  (p1 ph1):f1
  d6
  (p2 ph2):f1
;   d7
  go=2 ph31
  10m wr #0
  lo to 1 times l10
; 10m mc #0 to 1 F0(ze)
exit

ph1= 0
ph2= 0 1 2 3
ph31=0 2
```

- pulse flip angles and delays are somewhat arbitrary
- prescription for the optimum pulses and delays might not be given explicitly
- highest possible coherence order is equal to number of coupled spins multiplied by 2*S*

General rules for phase cycling - J. Magn. Reson. 1984, 56, 418; 1984, 58, 370.

1. $p = 0 \rightarrow \ldots \rightarrow -1$

2. $\phi_{Rx} = -\sum_i \Delta p_i \phi_i = -1 \cdot (\Delta p_1 \phi_1 + \Delta p_2 \phi_2 + \cdots)$

3. $\Delta\phi_i = \frac{2\pi}{N}$ selects $\Delta p \pm n \cdot N$

- because of (1.), phase cycling is not necessary for at least one pulse
- groups of pulses can be phase cycled together
- because of (3.), the following pathway is also selected



- alternative phase cycles for spin-echo

```
ph1= 0 1 2 3
ph2= 0
ph31=0 3 2 1
```

Rotor-synchronization and finite pulse widths




```
;se

"d6=(1s/cnst31)-(p1/2)-(p2/2)"
;"d7=(1s/cnst31)-(p2/2)"

1 ze
2 d1
  3u pl1:f1
  (p1 ph1):f1
  d6
  (p2 ph2):f1
;  d7
  go=2 ph31
  10m wr #0
  lo to 1 times l10
exit

ph1= 0
ph2= 0 1 2 3
ph31=0 2
```

- the duration of pulses have to be taken into account for delays that need to be synchronized with the sample spinning frequency

Two-dimensional experiments




```
;noesy, pdsd, chemical exchange

"l1=td1/2"

1 ze
2 d1
  3u pl1:f1
  (p1 ph1):f1
  d0
  (p1 ph2):f1
  d4
  (p1 ph3):f1
  go=2 ph31
  10m wr #0 if #0 zd
    10u ip1
  lo to 2 times 2
    5u rp1
    5u id0
  lo to 2 times l1
; 10m mc #0 to 2 F1PH(ip1,id0)
; 10m mc #0 to 2 F1PH(calph(ph1,+90), caldel(d0,+in0))
exit

ph1= 0 2
ph2= 0
ph3= 0 0 1 1 2 2 3 3
ph31=0 2 1 3 2 0 3 1
```

Hypercomplex acquisition

one-dimensional signal

$s(t) = \exp(i\Omega t)\exp(-t/T_2)$ — **FT** → $s(\omega) = A(\omega) + iD(\omega)$



phase modulated two-dimensional signal (black pathway only during t_1)

$s(t_1, t_2) = \exp(-i\Omega t_1)\exp(i\Omega t_2)$ — **FT₂** + **FT₁** →

$s(\omega_1, \omega_2) = (A_{1-} - iD_{1-})(A_2 + iD_2)$

$\mathrm{Re}[s(\omega_1, \omega_2)] = A_{1-}A_2 + D_{1-}D_2$



amplitude modulated two-dimensional signal (black+red pathway during t_1)

$s(t_1, t_2) = [\exp(+i\Omega t_1) + \exp(-i\Omega t_1)]\exp(i\Omega t_2)$ — **FT₂** + **FT₁** →

$s(\omega_1, \omega_2) = [(A_{1+} + iD_{1+}) + (A_{1-} - iD_{1-})](A_2 + iD_2)$

$\mathrm{Re}[s(\omega_1, \omega_2)] = (A_{1+} + A_{1-})A_2$



$$\exp(\pm i\theta) = \cos\theta \pm i\sin\theta$$
$$\exp(+i\theta) + \exp(-i\theta) = 2\cos\theta$$
$$\exp(+i\theta) - \exp(-i\theta) = 2i\sin\theta$$

we have: $s(t_1, t_2) = \cos(\Omega t_1)\exp(i\Omega t_2)$
we need: $s(t_1, t_2) = \sin(\Omega t_1)\exp(i\Omega t_2)$

$$\Delta\phi_i^{hypercomplex} = \frac{\pi}{2|p|}$$

States-Haberkorn-Ruben - J. Magn. Reson. 1982, 48, 286.

- repeat each t_1 point with phase of pulse before t_1 delay incremented by $\pi/(2|p|)$

$s_{\cos}(t_1, t_2) = \cos(\Omega t_1)\exp(i\Omega t_2)$ $\quad$ $s_{\sin}(t_1, t_2) = \sin(\Omega t_1)\exp(i\Omega t_2)$

— $\mathbf{FT_2}$ →

$s_{\cos}(t_1, \omega_2) = \cos(\Omega t_1)(A_2 + iD_2)$ $\quad$ $s_{\sin}(t_1, \omega_2) = \sin(\Omega t_1)(A_2 + iD_2)$

$s_{\text{States}}(t_1, \omega_2) = \text{Re}[s_{\cos}] + i \cdot \text{Re}[s_{\sin}] = \exp(i\Omega t_1)A_2$

— $\mathbf{FT_1}$ →

$\text{Re}[s_{\text{States}}(\omega_1, \omega_2)] = A_{1+}A_2$

```
"l1=td1/2"

1 ze
2 d1
  ...
  10m wr #0 if #0 zd
    10u ip1
  lo to 2 times 2
    5u rp1
    5u id0
  lo to 2 times l1
```

Time proportional phase incrementation - Faraday Symp. Chem. Soc. 1979, 13, 49.

- halve t_1 increment and increase phase of pulse before t_1 delay by $\pi/(2|p|)$ for every t_1 point

$s(t_1, t_2) = \cos\left[\left(\Omega + 2\pi\frac{\text{SW}_1}{2}\right)t_1\right]\exp(i\Omega t_2)$ — $\mathbf{FT_2} + \mathbf{FT_1}$ →

$\text{Re}[s_{\text{TPPI}}(\omega_1, \omega_2)] = \left[A_1\left(+\Omega + 2\pi\frac{\text{SW}_1}{2}\right) + A_1\left(-\Omega - 2\pi\frac{\text{SW}_1}{2}\right)\right]A_2$

```
"in0=inf1/2"

1 ze
2 d1
  ...
  10m wr #0 if #0 zd
    5u ip1
    5u id0
  lo to 2 times td1
```